

```
;-----  
; Programa: PWM2.asm  
; Controle PWM para motores DC  
; Autor    : Marcio Jose Soares  
; Data     : 01/10/2002  
;-----  
radix dec          ;padrão -> valores decimal  
include "P16F628a.inc"  
__config          0x3F29  
  
PICRAM equ 0x0C    ;endereço da RAM  
MOTOR  equ 0       ;Motor  
CHV1   equ 1       ;Chave para aumentar a velocidade do motor  
CHV2   equ 2       ;Chave para diminuir a velocidade do motor  
LED    equ 3       ;LED  
MAX    equ 0x59    ;valor máximo para PWM 01011001  
MIN    equ 0x0C    ;valor mínimo para PWM  
  
cblock 0x20  
    T1      ;variável para tempo  
    T2      ;variável para tempo  
    T3
```

```
        TMAX      ;variável para trabalhar tempo máximo
        TMIN      ;variável para trabalhar tempo mínimo
endc
```

```
org      0x000
goto     inicio      ;desvia do endereço 0x04 - interrupção
```

```
org      0x004      ;toda interrupção aponta para este endereço
retfie
```

```
;*****
;inicio do programa
;*****
```

```
inicio:
```

```
    bcf      INTCON,GIE
    clrf     PORTA
    movlw    0x07
    movwf    CMCON
    clrf     PORTB
    bsf      STATUS,RP0      ;seleciona BANK1
    movlw    b'11000110'
    movwf    TRISA
```

```
    clrf    TRISB
    bcf     STATUS,RP0      ;seleciona BANK0
    movlw   MAX             ;carrega TMAX -> 70 ms
    movwf   TMAX
    movlw   MIN             ;move valor para TMIN -> 10 ms
    movwf   TMIN

;*****
;Principal
;*****
loop:
    call    TECLA           ;verifica Tecla

PWM:
    bcf     PORTA,MOTOR     ;liga pulso
    call    TMINIM          ;aguarda
    bsf     PORTA,MOTOR     ;desliga pulso
    call    TMAXIM          ;aguarda
    goto    loop            ;loop !!

;*****
;subrotina para testar tecla
```

```
;*****
```

```
TECLA:
```

```
    nop                ;perde 3 ciclos
```

```
    nop
```

```
    nop
```

```
    btfss    PORTA,CHV2    ;testa chave 2
```

```
    goto     TC2           ;pressionada
```

```
    nop                ;perde 3 ciclos
```

```
    nop
```

```
    nop
```

```
    btfss    PORTA,CHV1    ;testa chave 1
```

```
    goto     TC1           ;pressionada
```

```
    goto     TC3           ;nenhuma chave pressionada
```

```
;*****
```

```
;subrotina incrementa velocidade
```

```
;*****
```

```
TC1:
```

```
    nop                ;perde 3 ciclos
```

```
    nop
```

```
    nop
```

```
    movf      TMIN,W           ;carrega com valor mínimo atual
    andlw     0xFF             ;faz and com W
    xorlw     MAX              ;faz xor com W, se = 1, bit Z em STATUS = 1
    btfss     STATUS,Z         ;testa resultado, zero valor mínimo = máximo
    goto      TC1_1            ;incrementa velocidade
    bsf       PORTA,LED        ;velocidade no máximo, não incrementa mais
    goto      TC3              ;e liga led

TC1_1:
    incf      TMIN,F           ;incrementa registro
    decf      TMAX,F           ;decrementa registro
    goto      TC3

;*****
;subrotina decrementa velocidade
;*****
TC2:
    nop                     ;perde 3 ciclos
    nop
    nop
    movf      TMIN,W           ;carrega com valor mínimo atual
    andlw     0xFF             ;faz and com W
```

```
        xorlw    MIN                ;faz xor com W, se = , bit Z em STATUS = 1

        btfss   STATUS,Z           ;testa se resultado zero
        goto    TC2_1              ;decrementa velocidade
        goto    TC3                ;velocidade no mínimo, não decrementa mai

TC2_1:
        decf    TMIN,F             ;incrementa registro
        incf    TMAX,F             ;decrementa registro
        bcf     PORTA,LED           ;desliga led

TC3:
        bcf     STATUS,Z           ;garante bit Z como 0
        return

;*****
;subrotina para temporização do PWM
;*****

TMINIM:
        movf    TMIN,W             ;carrega W com TMIN
        movwf   T1                 ;carrega T1 com W
```

```
    goto    tempo    ;inicia temporização
TMAXIM:
    movf    TMAX,W    ;carrega W com TMAX
    movwf   T1        ;carrega T1 com W

tempo:
    movlw   0xFF      ;carrega W com 255
    movwf   T2        ;carrega T2 com 255
again:
    decfsz  T2,1      ;decrementa T2
    goto    again     ;255 x T1 vezes
    decfsz  T1,1      ;decrementa T1
    goto    tempo     ;volta a carregar T2
    retlw   0x00

end
```